

LUNAR MODULE ATTITUDE CONTROLLER ASSEMBLY INPUT PROCESSING

Jose R. Portillo Lugo
jportillo34@hotmail.com

June 2004

INTRODUCTION

As a fully Electric vehicle, the Apollo Lunar Module provides the possibility for all its Guidance and Control functions to be managed throughout Automatic and Manual Control Paths. Operations like Attitude change can be commanded by an Electronic Autopilot or Manually by crewman Commands. The Crewman enters into the Control Loop, and in some cases into the Guidance Loop using the Attitude Controller Assembly or ACA. This Assembly provides a linkage from Crewman to the Lunar Module Electronic Control Systems.

There are three main Manual Control Paths implemented into the Lunar Module Electronic Control System. A primary Control Path, which process digitally all Manual Input Commands. A secondary or backup Control Path, which process all Manual Input Commands by Analog Electronic Networks. Additionally, a Direct or Auxiliary Control Path is provided, which bypass any Control Path.

This brief Monograph describes the Digital and Analog processing of Manual Inputs by the Apollo Lunar Module Primary Guidance, Navigation and Control System (PGNCS), and by the Abort Guidance System (AGS). Primary Control Path Digital processing of ACA Inputs is described in detail by use of a Digital Autopilot program, by use of Interface documentation and by use of background design information. Backup Control Path Analog processing of ACA Inputs is described and compared with similar Control systems. Pilot oppinions, extracted from real experiences are included as complement material.

The Monograph is structured in seven sections:

0. ACA ELECTRIC SOURCES AND ELECTRIC OUTPUT
1. PGNCS PROCESSING (Primary Control Path)
2. AGS PROCESSING (Secondary Control Path)
3. DIRECT MODE (2-Jets Path)
4. DIRECT MODE (Emergency 4-Jets Path)

REFERENCES AND SELECTED BIBLIOGRAPHY

APPENDIX

GLOSSARY

Descriptions related to Assembly Language Source Code Comments are enclosed in boxes (****). Also references to Panel Labels and Switch Labels are enclosed as <"SWITCH LABEL NAME"> or <"PANEL LABEL NAME">. Also, each Assembly Language line is numbered in sequence as "001".

The Glossary section provides definitions to basic concepts used in the text.

0. ACA ELECTRIC SOURCES AND ACA ELECTRIC OUTPUT

*There is a powerful agent, obedient, rapid, easy,
which conforms to every use, and reigns supreme
on board my vessel. Everything is done by means of it.
It lights, warms it, and is the soul of my mechanical apparatus.
This agent is electricity.*

- Jules Verne, 20,000 Leagues Under The Sea (1869-1870)

LUNAR MODULE ELECTRIC SOURCE

The Apollo Lunar Module is an Electric vehicle powered by seven Batteries and by a complete Electrical Power Subsystem, which includes a DC section and an AC section.

Two Descent stage Batteries power the Lunar Module from T-30 minutes (mission time) until after Transposition and Docking, at which time the Lunar Module receives Electrical power from the Command and Service Module. After separation from the Command and Service Module, during the powered descent Phase of the mission, four Descent Stage Batteries are paralleled with the Ascent Stage Batteries. Paralleling the Batteries ensures minimum required Voltage for all the Lunar Module operations. During Lunar surface stay, specific combinations of the five Descent Stage Batteries can be paralleled to provide Lunar Module Power. [3]

Before Lift-off from the lunar surface, Ascent Stage Battery Power is introduced, and Descent Power is terminated. Ascent Stage Battery Power is then used until after final Docking and Crew transfer to the Command Module. [3]

Electrical Power is distributed to the Lunar Module Subsystems via the Lunar Module Pilot's DC BUS (some references Refers to this Crewmember as Systems Engineer), the Commander's DC BUS, and AC BUSES A and B.

ACA GENERAL ARRANGEMENT

Within its design, the ACA has two kinds of Outputs for each Axis: a Proportional Output, and two Discrete (Switch type) Outputs, which send Discrete Electrical Signals to activate the Manual operation for several Control Modes within the LM Flight Control System Electronic Network (ATCA, LGC-ATCA).

ACA ELECTRIC SOURCE

ACA Electric Source comes from three different Paths:

- PSA and LGC in PGNS (Primary Path). In this Signal Path the ACA receive:
 - a) A 28-Volt DC Electric Signal from the LGC for Discrete Commands like MINIMUM IMPULSE and 2-Jets DIRECT Processing (see Control Modes on page 5).
 - b) A 28-Volt 800 cps AC Electric HI/LO Source from the PSA for Proportional Commands Processing (see "ACA Proportional Output" on page 3). [21]
- ATCA in the AGS (Secondary Path). In this Signal Path the ACA receive:
 - a) A 28-Volt DC Electric Signal from the ATCA for Discrete Commands Like PULSE and 2-Jets DIRECT Processing (see Control Modes on page 27).
 - b) A 28-Volt 800 cps AC Electric HI/LO Source from the ATCA for Proportional Commands Processing (see "ACA Proportional Output" on page 3). [21]

- EMERGENCY DIRECT (4-Jets Path). In this Signal Path the ACA receive:

- a) A 28-Volt DC power directly from the <"ATT DIR CONT"> Circuit Breaker in the <"STAB/CONT"> Breaker Panel Section.




Figure 1. The Attitude Controller Assembly (ACA)

ACA PROPORTIONAL OUTPUT

For ACA Proportional processing an ACA Handgrip movement produce a rotary displacement on the ACA RVDT-type Synchro Moving Core or Armature, and an AC Output Signal is generated with Voltage Amplitude proportional to the displacement. [2]

There are two Coils surrounding the Armature: the Excitation or Input Coil, and the Output Coil. This Coil/Armature configuration is repeated for each axis (Pitch, Roll, Yaw).

The 28-Volt AC 800 cps HI/LO Source Signal (the ACA Electrical source described on page 2) is applied to the primary Coil of the RVDT for the axis selected. When ACA displacement exceeds 0.5 Deg., a set of Switch contacts close (see figure 1). These Switch contacts send a Proportional secondary voltage to the LGC or to the ATCA. The output gradient voltages are 0.28 Volt per Degree of displacement of the Armature (1 Deg. to 11 Deg.). Positive rotation Commands generate in-phase voltage, and negative rotation commands generate 180 Deg. out-of-phase Voltage. [3] [28]

The 28-Volt AC 800 cps HI/LO Proportional Output goes to the LGC in the Primary Path, and to the ATCA in the Secondary Path.

NOTE:

The ACA is a Transformer based circuit. The LGC has a special Interface Circuit for ACA Proportional Input. Called the "A Circuit", It does an Analog-to-Digital conversion on the ACA Analog Proportional AC Input (see PGNS RATE-COMMAND-AUGMENTATION Mode section on page 6).




Figure 2. The ACA-LGC Proportional Interface ("A" Circuit at right)

In practice, the ACA handgrip must be deflected up to 1.5 Deg. more after Detent Switch close. At this point the output Voltage begins to build up. Having reached the Voltage ramp at 2 Deg. of deflection, the commanded Rate increases 2.5 Deg. per Sec. for each Deg. of handgrip deflection. [6]

For the ACA Proportional Output to be used, the <"ATCA (PGNS)"> or <"ATCA (AGS)"> Circuit Breakers in the <"STAB/CONT"> Breaker Panel section have to be closed, and the <"ACA PRO"> Switch have to be toggled to <"ENABLE"> position. The <"ACA PRO"> Switch enables the Electric Source for the ACA Primary Coil (Excitation or Input Coil). As described on page 2, Electric Sources comes either from the PSA in PGNS (Primary Path) or from the ATCA in AGS (Secondary Path), depends on the <"GUID CONT"> Switch position (PGNS position or AGS position).

1. PGNS PROCESSING (Primary Control Path)

Through the Primary Attitude Control Path, the crew can use one of four Control Modes for Attitude maneuvering (see Figure 2.1 for switch arrangement in Switching Panel):

- PGNS RATE-COMMAND-AUGMENTATION Mode

(<"MODE CONTROL"> Switch in <"ATT HOLD"> position)

The ACA Electric Proportional output enters the LGC "A" Interface Circuit for processing. The Attitude-Hold feature engages whenever the ACA Stick is returned to the Detent position.

This Mode can be entered by keying Verb 77 (V77E) in the DSKY User Interface.

The ACA Proportional Output shall be active through the entire ACA deflections (+/- 10 Deg.) except for the Dead Band at neutral position (see Figure 1). This is a Mechanical/Electrical Dead Band, which do not exceed +/- 1 Deg. ACA deflection.

The Proportional Output falls outside the Mechanical Dead Band to prevent inadvertent inputs. [12]

- PGNS MINIMUM IMPULSE Mode

 (<"MODE CONTROL"> Switch in <"ATT HOLD"> position)

This is an Open-Loop mode on which the crew can obtain a single impulse by moving the ACA Handgrip to 2.5 Deg. deflection or beyond and then returning it to detent position.

This mode can be entered by keying Verb 76 (V76E) in the DSKY User Interface. While in this Mode, a DC Excitation Voltage is applied to the ACA Plus and Minus Switches (2 per-axis) for PULSE and LPD (see the MINIMUM IMPULSE and LPD sections below). This DC Excitation Voltage comes as a Common Voltage from the 28-Volt DC Source of the LGC. The ACA Discrete Electric Output goes in return to the LGC. Hence, when the ACA Stick is displaced 2.5 Deg. or more from neutral, a positive or negative DC voltage is applied to the LGC Channel 31 Inbits (see MINIMUM IMPULSE section below).

The MINIMUM IMPULSE Mode is actuated at the same Handgrip displacement point as the DIRECT Mode.




Figure 2.1. Stabilization and Control Switching Panel

- 2-Jets DIRECT Mode

 (<"ATTITUDE CONTROL"> Switch in <"DIR"> position)

See section 3, DIRECT MODE (2-Jets Path).

- EMERGENCY DIRECT Mode

(4-Jets)

See section 4, DIRECT MODE (Emergency Path).

PGNS RATE-COMMAND-AUGMENTATION Mode

The RATE-COMMAND-AUGMENTATION Mode is structured as a combination of Human-Pilot Manual Input Processing and Automatic Rate-Damping. It is a Semi-automatic Mode by which the Crewman can interrupt the Automatic Guidance Loop and establish a new Attitude reference from which he (she) can return the System State to Automatic Guidance or take Manual Control. The name Semi-Automatic is related to the way by which the Manual intervention is located in the Guidance Loop: the Automatic System acquire the "Next Step" from its internal Guidance Instructions or Plan, but Crewman can enter Manually a new Plan (in this case the Automatic System have to have an Interface within its Control Logic to permit Manual Inputs).

This Mode is the normal way by which the Astronauts Control the LM. The Mode is divided in two sections: one Manual, and another one Automatic. The Manual section is what is called RATE-COMMAND and it provides Rate Command capability feedback controlled through a Rate Error Dead Band when the ACA is out of the Detent position. The product of processing at this point is a Rate Error Signal that commands the Electronic Control section of the LM System. The Automatic section enters into action whenever the ACA is in Detent position, and the Rate Error about every Axis has been reduced to a small (established) value (a Process called Rate Damping, that in the case of LUMINARY Program the return to Automatic Mode is forced if Damping has not been completed by the end of a brief interval).

To accomodate the various data formats, the interfaces were designed around the logical operations provided by the incrementing and shifting capabilities of the (LGC) counters and the capability of reading and writing into the (LGC) channels. [4]






Figure 3. The LM Guidance Computer (LGC) and the "A" Circuit
("D" Circuit is located in the same Interface Modules)

At the ACA Interface level, the LGC has a special Interface Circuit for ACA Proportional Input that is used for RATE-COMMAND Mode Inputs. Called the "A Circuit", this Interface Circuit, along with the LGC Counter system, executes an Analog-to-Digital conversion on the ACA Analog Proportional AC Input and produces a Digital Binary representation in one of the LGC RHC (Rotational Hand Controller - the ACA) Interface Counters: one Counter per-Axis. [4]

The "A Circuit" internal arrangement consisting of Transistor Switching Logic for gating processing, and a Transformer based AC Analog Circuitry (see Figure 3 and page 28).

Every time the ACA is in the Out of Detent position, in one of the Control Axes, the Switching Process within the "A Circuit" produces a Signal (bit) with Low ("0") state by a time Proportional to the Amplitude (AM) of the ACA Handgrip displacement Signal. This output bit (on low "0" state) gates a 3,200 pps pulse train, which is an input to a LGC Counter (Q-RHCCTR, or P-RHCCTR, or R-RHCCTR). The resulting Binary number in the Counter is a Digital representation Proportional to the ACA displacement (see Figure 4). [3] [4] [5]

Each one of the LGC Counters (Q-RHCCTR, P-RHCCTR, or R-RHCCTR), is updated by its correspondent Counter Interrupt.

The ACA Output Voltage is digitized to 42 counts at the 10 Deg. soft stop deflection (soft stop deflection is considered full-scale, although voltage and counts increase until hard stop is reached at 13 Deg. Deflection).

This Analog-to-Digital conversion technique is called "Incremental Information Transfer" and was adopted as a means of Analog data Transmission in order to maintain high precision with relatively simple techniques of Analog-to-Digital conversion. [4]

At this point in the ACA Input Processing, the digitized amount of ACA Deflection is ready for Digital Processing inside the LGC. The digitized quantity is Scaled to a Rate Command value before it is ready to be used by the RATE-COMMAND Software.

The Signal/Data Path of the ACA Digital Processing is structured in 5 Sections:

- RATE-COMMAND INITIALIZATION
- ACA INPUT RATE SCALING
- DIRECT RATE/PSEUDO-AUTO SWITCHING LOGIC
- DIRECT RATE (JET FIRING COMMAND) MODE
- PSEUDO-AUTO MODE

RATE-COMMAND-AUGMENTATION MODE INITIALIZATION

The LM Primary Control Path (AUTOMATIC/ATTITUDE-HOLD and RATE-COMMAND) is implemented in a Software Component called the Digital Autopilot or DAP. The RATE-COMMAND is available for the Crew each time a Verb 77 (V77E) is entered in the LGC DSKY Interface. While in RATE-COMMAND Mode, the maximum maneuver Rate about any Axis of the LM is 20 Deg./Sec. [27]

The LM Digital Autopilot is a first generation Digital Control System that would have been difficult to implement with conventional Analog Autopilot design techniques. [26]

The main benefit inherent in Mechanizing Controller Logic in a Digital Computer like the LGC is the flexibility to perform complicated Control-System Logic with essentially no Hardware impact.

Every 10 times per-second, the Guidance Software being processed in foreground (Automatic Guidance Programs) in the LGC is interrupted. Then, the LGC internal State Vector is stored in the Erasable Memory, and the Control Sequence is transferred to the DAP Program.

The LGC is organized such as Input/Output Interfaces are attended by a Central Processor on a time-shared basis with other Guidance processing functions. Input Signals as ACA Input are converted to Digital Data by use of a Counter Interrupt Procedure, which is assigned the highests priority in the LGC time-sharing sequence.

NOTE:

According to Professors Patterson and Hennessy in reference 25, an Interrupt is " an unscheduled procedure call. The address of the (next) instruction is saved in a register, and the computer jumps to a predefined address to invoke the appropriate routine for that exception (Interrupt). " [25]

(see Appendix A)

As in every Control System, the DAP works over an established Parameters Framework. The Framework includes: information about the Spacecraft Mass, Dead Bands, Maximum Commanded Rate, and input Command Scaling.

Extended VERB 48, the Digital Autopilot Parameters loading Routine, is used to enter those Control System Parameters.

The DAP DATA CODE (N46), which includes all the DAP initialization Parameters has 5 digits. Among them are the ACA Scaling Parameter, and the Dead Band Parameter:

DIGIT C - ACA Scaling

0 - FINE (4 Deg./Sec.)
1 - NORMAL (20 Deg./Sec.)

DIGIT D - DEADBAND

0 - 0.3 Deg.
1 - 1.0 Deg.
2 - 5.0 Deg.
(3 - 5.0 Deg. by default)

DAP Assembly Language Source Code for "STICKLOAD":

```
*****  
* Variables and Constants used in this section: *  
* * * * *  
* -0.6D/S ---- LM-ONLY Breakout Level; Decimal value is -218 *  
* -RATEDB ---- Established Breakout Level for the Switchng Function (-0.6 or -0.3) *  
* NORMAL ----- .660214 (4 Deg./sec.) *  
* FINE ----- .165054 (20 Deg./Sec.) *  
* TARGETDB ---- Manual Mode Target Dead Band. It is set equal to -RATEDB *  
* STICKSENS --- ACA Scaling; Maximum Commanded Rate (FINE, NORMAL) *  
* DAPBOOLS ---- DAP Flagword where all the DAP configuration is stored *  
* RHCSALE ---- Field of DAPBOOLS where the ACA Scale is stored *  
*****  
001 STICKLOAD CAF EBANK6  
002 TS EBANK  
003 EBANK= STICKSENS  
004 CA RHCSALE Set Stick sensitivity to correspond to a  
005 MASK DAPBOOLS Maximum Commanded Rate (at 42 counts) of  
006 CCS A 20 Deg./sec. (NORMAL) or 4 Deg./Sec. (FINE),  
007 CA NORMAL Scaled at 45 Deg./Sec.  
008 AD FINE  
009 TS STIKSENS  
010 CA -0.6D/S  
011 TS -RATEDB LM-ONLY breakout level is .6 Deg./Sec.  
012 CA CSMDOCKD If CSM-DOCKED, divide Stick Sensitivity  
013 MASK DAPBOOLS by 10. NORMAL scaling is then 2 Deg./Sec.  
014 EXTEND and FINE scaling is 0.4 Deg./Sec.  
015 BZF +7 Branch if CSM is not Docked  
016 CA STIKSENS  
017 EXTEND  
018 MP 1/10  
019 TS STIKSENS  
020 CA -0.3D/S CSM-DOCKED breakout level is .3 Deg./Sec.  
021 TS -RATEDB  
022 RELINT Proceed to noun 47, Mass Load
```

The TIME5 Counter or DAP Counter is one of six Counters used for timing purposes in the LGC. This Counter establishes the DAP time interval (0,10 Seg.). Before DAP computations reach its completion (iteration), TIME5 Counter is reset so that it will overflow at a specified time in the future.

The TIME5 Counter is incremented on address 00030 (Octal) until overflow (DAP processing). The Overflow on this Counter generates an Interrupt "T5RUPT" on address 4010 (Octal). [3]

The Control is then transferred to the "P Axis" label on the Luminary program (DAP).

Into the "P Axis" section: bit 15 on input Channel 31 (Out of Detent) is verified:

```
- IF bit 15 = "1" THEN
  {
    Transfer control to Label "RHCMOVED"
  }
```

DAP Assembly Language Source Code for "DETENTCK":

```
*****
* Variables and Constants used in this section: *
*
* OURCBIT ----- A Flag bit for indication DAP State was in Rate-Command during last DAP cycle *
* RCSFLAGS ---- DAP Flag word (includes bits like PBIT) *
* CHAN31 ----- Channel 31 *
* PBIT ----- A Flag bit for requesting the use of DIRECT RATE Mode during next DAP cycle *
*****
023   DETENTCK   EXTEND
024           READ  CHAN31
025           TS    CH31TEMP
026           MASK  BIT15      Check Out-Of-Detent bit.
027           EXTEND
028           BZF   RHCMOVED   Branch if Out of Detent
029           CAF   OURRCBIT   In Detent --> Check the Rate Command bit.
030           MASK  DAPBOOLS
031           EXTEND
032           BZF   PURGENCY   Branch if not in Rate Command last pass.
032.1         CA    BIT9      Check JUSTIN bit
032.2         MASK  RCSFLAGS
032.3         EXTEND
032.4         BZF   RUTH
032.5         CAF   BIT13     Check Mode for ATTITUDE HOLD
032.6         EXTEND
032.7         RAND  CHAN31
032.8         EXTEND
032.9         BZF   RATEDAMP   Branch to Rate Damp if in ATTITUDE HOLD (Source Line 059)
032.10        CS    BITS9,11  In AUTO (X-Axis Override)
032.11        MASK  RCSFLAGS
032.12        TS    RCSFLAGS   Zero "ORBIT" and Just-In Bit
032.13        TCF   RATEDAMP
032.14 RUTH     CA    RCSFLAGS
032.15        MASK  PBIT      In ATTITUDE HOLD
032.16        EXTEND
032.17        BZF   +2        Branch if P Rate Damping is finished
032.18        TCF   RATEDAMP
```

Code on Label "RHCMOVED" initialize ACA input Counters, and sets output bits 8 and 9 (Start RHC read, and RHC Counter enable) in Channel 13. Bit 8, is equivalent to Pin "A" on the AGC "A Circuit". [3] [5]

NOTE:

The two-step Processing of ACA Proportional Input permits to close the Out-Of-Detent Discrete without Commanding a Rate as stated in Reference 6 (see bits 8,9 use on page 10 and comment on page 8 of this document).

DAP Assembly Language Source Code for "RHCMOVED":

```

*****
* Executive Control of the Rate-Command Mode, including initialization, reading, *
* zeroing, and enabling of ACA Counters, and the decision to continue Rate Damping *
* or to return to Automatic Control, occurs in the P-Axis Coding (Program section) *
* Ref. [5] *
*****
033          TS      PERROR
034  JUSTOUT  CA      OURRCBIT   Initialization - First Manual Pass.
035          ADS     DAPBOOLS
036          CA      ZERO
037          TS      DXERROR
038          TS      DXERROR + 1
039          TS      DYERROR
040          TS      DYERROR + 1
041          ....
042          TS      Q-RHCCTR
043          TS      R-RHCCTR
044  ZEROENBL LXCH    R-RHCCTR
045          CA      Q-RHCCTR
046          DXCH    SAVEHAND
047          CA      ZERO
048          TS      P-RHCCTR
049          TS      Q-RHCCTR
050          TS      R-RHCCTR

```




Figure 4. LGC Input Counters and its corresponding ACA Axes

```

*****
* This section activates the "A" Circuit Gating Process by setting bit 9 on Channel 13. *
* bit 8 is activated to enable the RHCCTR Counter Increment (Decrement) for the active Axis. *
* *
* Constant used: *
* *
* BITS8,9 = 00600 (Octal value) *
*****
051          INHINT      Disable Interruptions.
052          EXTEND
053          QXCH  C13QSAV
054          TC      C13STALL   Clear bits
055          CA      BITS8,9
056          EXTEND
057          WOR     CHAN13     Counters Zeroed and Enabled.

```

```

058          TC      C13STALL
059  RATEDAMP  CA      ZERO
060          TS      P-RHCCTR  <-- Is processing P-Axis
061          TCF     RATERROR
062  RHCMOVED  CS      RCSFLAGS  Set JUSTIN bit to 1.
063          MASK   BIT9
064          ADS    RCSFLAGS
065          CA      OURRCBIT  P Control.
066          MASK   DAPBOOLS
067          EXTEND
068          BZF    JUSTOUT - 1
069  RATERROR  CA      CDUX      Find CDUX requires That CDUXD = CDUX
070          TS      CDUXD     during X-Axis override.

```

ACA INPUT RATE SCALING

The next step in the program is the digital processing of ACA proportional input signal. It is scaled to a quadratic function (Wc).

Asked about the use of scaling functions as applied to the ACA Input, Professor Robert Stengel (LGC Manual Rate-Command Mode designer) wrote:

" the scaling described is the sort of thing that was required when programming with single-precision, fixed-point numbers. One must be concerned about the granularity (or resolution) as well as the range of values, not only of the original numbers but of their sums and products. "
[8]

Quadratic Scaling of the ACA Input Signal to the LGC minimizes the difficulty in commanding small Rates, retaining high Command Rate capability. [6]

The LM-3 (Apollo 9) SUNDANCE Program Scaled the ACA Counter Input to a Linear Function, which is the usual way for old Analog Rate-Command systems design. Its Linear Scaling Law for Commanded Rate was established in Reference [22] as:

$$Wc = K1 \left\{ \frac{[7.15E]}{VRMS} \right\} \quad (1)$$

Where K1 = 1 for NORMAL ACA Scaling
K1 = 0.2 for FINE ACA Scaling
E = ACA Voltage in Volts RMS
Wc = Commanded Rate in Deg./Sec. [22]

In order to adapt the Maximum Commanded Rate (MCR) to cover emergency conditions that require the Urgent Commands response of the DIRECT RATE Mode, the LUMINARY Program incorporated a Linear-Quadratic Scaling Law for Commanded Rate (Figure 5). It was established in Reference [22] as:

$$Wc = K2 \left\{ \frac{[1.43E]}{V \text{ RMS}} + \frac{[2.04E]}{V \text{ RMS}} \right\} \quad (2)$$

Where K2 = 1 for NORMAL ACA Scaling
K2 = 0.2 for FINE ACA Scaling
E = ACA Voltage in Volts RMS
Wc = Commanded Rate in Deg./Sec. [22]

Applying Horner's Rule for Polynomial evaluation to equation 2:

$$Wc = K2 ((1.43 + 2.04E) E) \quad (3)$$

In Reference [6], Professor Stengel established the Luminary Commanded Rate Scaling as (equ. 4 in Ref. 6):

$$W_c = MCR (0.004760175E + 0.00045335E^2) \quad (4)$$

And applying Horner's Rule to equation 4:

$$W_c = MCR ((0.004760175 + 0.00045335E) E) \quad (5)$$

Where MCR is Maximum Commanded Rate (same as K1 and K2 above)

NOTE:

The software Scaling process uses the Input Counter value instead of the "E" ACA Voltage Value.




Figure 5. ACA Quadratic Scaling (reference 6)

DAP Assembly Language Source Code for ACA Scaling:

```

*****
* Variables and Constants used in this section:
*
* STICKSENS --- ACA Scaling; Maximum Commanded Rate (FINE, NORMAL)
* P-RHCCTR ---- ACA Stick Deflection stored in the LGC Counter
* DAPTEMP1 ---- P-Axis ACA Commanded Rate (Scaled in this section)
* LINRATP ----- Linear/Quadratic Sensitivity (B in equ. 6 below)
* PLAST ----- Commanded Rate calculated during last DAP Cycle
*****
071      CCS      P-RHCCTR      D (ACA Stick Deflection)
072      TCF      +3
073      TCF      +2
074      TCF      +1
075      DOUBLE
076      DOUBLE      Linear/Quadratic Controller Scaling Calculation
077      AD      LINRATP      |D| + B
078      EXTEND
079      MP      P-RHCCTR      (|D| + B) * D
080      CA      L
081      EXTEND
082      MP      STICKSENS      (|D| + B) * D) * A
083      XCH      PLAST

```

```

084          COM
085          AD    PLAST
086          TS    DAPTEMP1    P-Axis ACA Commanded Rate

```

This ACA Scaling Processing Source Code is the linearized equation ground to reality by the LGC Assembly Language:

$$Wc = A * (B + |D|) * D \quad (6)$$

Where

```

Wc = Commanded Rotational Rate
A  = Quadratic Sensitivity Factor
B  = Linear/quadratic Sensitivity
|D| = Absolute Value of ACA Stick Deflection
D  = ACA Stick Deflection [5]

```

About the quadratic function, Professor Robert Stengel wrote:

" The astronauts preferred to have fine precision for small commands, as well as high maximum commanded rate. This could have been accomplished with two linear segments of varying slope; however, the astronauts could discern the change in slope and were uncomfortable with it.

The quadratic function provided a simple, smooth solution to the problem. " [8]

Simulation work done in the LM Mission Simulator during 1968 (see reference 6), using the Luminary Manual Rate Command Logic, discovered an improvement in handling qualities by reducing the MCR Parameter. However, as described by Professor Stengel, reduced sensitivity made small rates and small angle changes easier to obtain, but there was concern that the MCR was insufficient for emergency conditions. A 20 Deg./Sec. MCR was deemed mandatory by the Astronauts. So, the solution that has been adopted is nonlinear Scaling of the ACA Output.

The FINE Scaling (4 Deg./Sec.) could be configured (as a DAP Configuration Parameter with VER 48) for LM-CSM Docking tasks, while the NORMAL Scaling (20 Deg./Sec.) could be configured for Lunar Landing tasks.

NOTE:

Previous to 1968, lunar landings were reproduced and presented in an AIAA conference in 1967 (see reference 34) showed that there was a large range of System Control Parameters, which produced acceptable handling qualities when using a Rate-Command-Augmentation system and also indicated that optimum values for Control Power and Maximum Commanded Rate were about 12.5 Deg/Sec². and 12.5 Deg./Sec. respectively.

The Simulation work was conducted at the Lunar Landing Research Facility located at the Nasa Langley Center: a Tethered test Vehicle suspended or supported by an electromechanical Infrastructure, which allowed to experience with several Control System configurations for Manual Lunar Landing Operations (Augmented or not) and with different combinations of basic Control Parameters as Dead Bands, Maximum Commanded rate, and Control Power, and these all in an artificially designed pseudo-lunar gravity environment. In that research tool, the Rate-Command-Augmentation system logic provided rate command within a simple Rate Error Dead Band when the Hand-Controller was Out-Of-Detent: no one of the Luminary program Digital processing refinements (as described here) were present.

DIRECT RATE/PSEUDO-AUTO SWITCHING LOGIC

In the case of the RATE-COMMAND Mode for Program Luminary, two Modes of Control are employed: The DIRECT RATE Mode and the PSEUDO-AUTO Mode.

In the DIRECT RATE Mode, as described in the Luminary program source listing:

" RCS Jets are fired when ACA Handgrip position changes by a fixed number of increments in one DAP cycle. " [5]

In the PSEUDO-AUTO Mode, as described in the Luminary program source listing:

" Body-Fixed Rate and Attitude Errors are supplied to the TJETLAW, which exercises control. " [5]

NOTE:

For a description of the TJETLAW subroutine, see section "PSEUDO-AUTO MODE".

The selection of one of these two Modes is done by a Switching Function programmed within the Rate-Command Control Logic of the Luminary Program: the ACA Handgrip deflection is Scaled to a Rate Command (see "RATE SCALING" section), which is differenced with the Estimated LM Rate to form a Rate Error. Also, the same Rate Command (ACA Scaled) is Integrated and Subtracted from an Integrated Sum of IMU Gimbal Angle Increments transformed to Body Axes, providing an Attitude Error. Attitude Error and Rate Error are referenced to Body Axes.

The Commanded Rate (ACA Scaled) is then used by the Control Logic (RATE-COMMAND Mode) to perform the Switching Function: if the difference between the actual Commanded Rate and the previous value of the Commanded Rate (calculated during the last DAP cycle) exceeds the Breakout Level (a DAP parameter, see "-RATEDB" variable at source line 010), the DIRECT RATE Mode is used.

```
- IF "Change of Wc in two successive DAP cycles" >= "Breakout Level" THEN
  {
    Ew = OMEGAP - Wc * Use DIRECT RATE Mode to null Rate Error *
      -[Ew]
    Tjet = ----- * and Compute a Firing Time (Tjet) *          (7)
           a
  }
```

Where:

Tjet = Firing Time for the RCS Thrusters

Wc = Commanded Rate * from equation 6; also source line 086 *

OMEGAP = Estimated P-Axis Rate * source line 089 *

Ew = Rate Error = OMEGAP - Wc * Rate Error is used by either Direct Rate or Pseudo-Auto *

a = Net Rotational Acceleration

Breakout Level = 0.6 Deg./Sec. (0.3 Deg./Sec.) * variable "-RATEDB" at source line 010 *

The sign of Tjet determines the sense of the commanded rotation.

DAP Assembly Language Source Code for the Manual Rate-Command Switching Function:

```
*****
* This section of the Source Code is only for the P-Axis RCS Autopilot processing. *
* Similar Rate Error section is included in the Autopilot Code for the Q, R Axes. *
*                                                                                   *
* Variables used in this section:                                                                                   *
*                                                                                   *
* OMEGAP ----- Estimated P-Axis Rate                                                                                   *
* EDOTP ----- Rate Error in P-Axis ("Ew" in equation 7)                                                                 *
* DAPTEMP1 ---- P-Axis ACA Commanded Rate ("Wc" in equation 6)                                                           *
* -RATEDB ----- Established Breakout Level for the Switchng Function (-0.6 or -0.3) *
* TARGETDB ---- Manual Mode Target Dead Band (initially set as -RATEDB)                                               *
* PBIT ----- Manual Control Switch bit in RCSFLAGS Eresable Memory word                                             *
* RCSFLAGS ---- Permanent Autopilot Flag Word                                                                           *
```

```

* TJP ----- Jet Time ("Tjet" in equation 7)
*****
087      RELINT
          *****
          * COMPUTE RATE ERROR (Ew) *
          *****
088      CS   PLAST      Commanded Rate (one's complemented for subtraction)
089      AD   OMEGAP     Estimated P-Axis Rate
090      TS   EDOTP      Ew = (Estimated P-Axis Rate - Commanded Rate)

          *****
          * This is the DIRECT RATE Switching Function *
          *****
091      CCS  DAPTEMP1   If P Command change Exceeds Breakout Level
092      TCF  +3         Go to DIRECT RATE Control
093      TCF  +8         Not, Then check for DIRECT RATE Control last time
094      TCF  +1

          *****
          * This Breakout Level is set at STICKLOAD Source Code *
          *****
095      AD   -RATEDB
096      EXTEND
097      BZMF +4
098      CA   40CYC
099      TS   TCP
100      TC   PEGI
101      CA   RCSFLAGS   Check for DIRECT RATE-COMMAND last time
102      MASK PBIT
103      EXTEND
104      BZF  +2
105      TC   PEGI       to DIRECT RATE-COMMAND Processing
106      CA   DXERROR    PSEUDO-AUTO Control
107      TS   E          X-ATTITUDE ERROR
108      TS   PERROR     Load P-AXIS ERROR for MODE1 FDAI DISPLAY
109      TC   PURGENCY +4

```

DIRECT RATE (JET FIRING COMMAND) MODE

The DIRECT RATE Mode computes Jet-ON time based only in Rate Error computation and doing it very fast.

In this case, if the Firing Time Computed (Tjet) exceeds 150 MSec., the Jets are commanded ON, and a new firing Time is computed on the next DAP Cycle. If the Firing Time computed is less than 150 Msec. (but not zero), then the Jets are commanded ON for the correct time, but the Computation is skipped on the next (DAP) Cycle. This Procedure takes place: a) every time the Breakout Level (0.6 Deg./Sec. or 0.3 Deg./Sec.) is exceeded, b) until the Rate Error falls within the Target Dead Band (before pass to Rate Damping), or c) until a time limit is exceeded. [5]

DAP Assembly Language Source Code for DIRECT RATE Mode:

```

110      PEGI      CA   CDUX      DIRECT RATE Control
111              TS   CDUXD     CDU desired
112              CA   ZERO
113              TS   DXERROR
114              TS   DXERROR +1
115              TS   PERROR     Zero P-Axis Error for MODE1 FDAI DISPLAY
116              CCS  EDOTP      P-Axis Rate Error (Ew)
117              TC   +3
118              TS   ABSEDOTP
119              AD   TARGETDB
120              EXTEND          If RATE ERROR is less than DEADBAND Then
121              BZMF LAST      Fire and then Switch to PSEUDO-AUTO
122              CA   TCP
123              EXTEND          If Time in RATE-COMMAND Exceeds 4 Seconds Then

```

```

124          BZMF  LAST      Switch to PSEUDO-AUTO
125          CS    RCSFLAGS
126          MASK  PBIT
127          ADS   RCSFLAGS  bit 10 is 1
128          TCF   +4

*****
* Fire Jets for Null Rate Error (Rate Damping) *
*****

129  LAST      CS    PBIT
130          MASK  RCSFLAGS
131          TS    RCSFLAGS  bit 10 is 0

*****
* Computing Tjet = -[Ew]/a ; a = Rotational Acceleration *
*****

132          CS    EDOTP      One's complement of Ew (-[Ew])
133          EXTEND
134          MP    1/ANETP     ½ JTACC Scaled at 2EXP(7)/PI
135          DAS   A
          .....
136          MP    25/32      Register A Contains Tjet Scaled
137          TS    TJP        Jet Time (Tjet)
137.1        CA    ABSEDOTP
137.2        AD    -2JETLIM   Comparing Delta Rate with 2 Jet Limit
137.3        EXTEND
137.4        BZMF  +3
137.5        CA    SIX
137.6        TCF  +80
137.7        CA    TJP
137.8        ADS   TJP

*****
* P-Jet Selection Routine *
* * * * *
* Input: TJP (+ signed for +P Rotation) *
*****

138          CA    AORBSYST
139          MASK  FLAGWRD5
140          CCS   A
141          CA    ONE
142          AD    FOUR
143          TS    NUMBERT
144  PJETSLEC CA    ONE
145          TS    L
146          CCS   TJP        Jet Time (Tjet)
147          TCF   +5
148          TCF   JETSOFF
149          TCF   +2
150          TCF   JETSOFF
151          ZL
152          AD    ONE
153          TS    ABSTJ
154          LXCH  ROTINDEX
155          TC    SELECTP     Call Subroutine to select a sub-set of Jets
156          CS    SIX
157          AD    NUMBERT
158          EXTEND
159          BZF   +2
160          CS    TWO
161          AD    FOUR
162          TS    NO.PJETS
163          CA    POLYTEMP
164          TC    WRITEP      in WRITEP Subroutine RCS Jet output bit at Channel 6 is set

```

```

*****
* If Tjet < 150Msec. Then Go to Q, R Axes Processing *
*****
165      CS   ABSTJ
166      AD   +150MST6
167      EXTEND
168      BZMF  QRAXIS

```

From this last Assembly Instruction the DAP Code follows a conditional path until Jet Command Output is executed by way of LGC Output Channel 5 (for P-axis): the Jet Time is stored in variable "T6FURTHA" for execution by the Waitlist Task "JTLST" during the Interrupt 6 (T6RUPT).

The DIRECT RATE Mode has two important attributes: it provides immediate response to (ACA) urgent Commands, while ignoring slow ACA inputs. The commanded Rate need not be large (ACA Handgrip deflection), but if it is requested quickly, the DIRECT RATE Mode is used. [5]

PSEUDO-AUTO MODE

This Mode stabilizes the LM about each of the IMU's reference Axes to Hold the Vehicle to within a specified Dead Band. Vehicle Attitude is used as input: Attitude cannot drift beyond the Attitude Dead Band, and small Rate Errors are Integrated to the Coast Zone Boundary, where corrective RCS Firing is Commanded. It limits drift about uncommanded Axes and makes precision Rate Control possible. It is possible to obtain Commanded Rates as small as the Quantization Level of the ACA.

If the Rate Command (ACA Input) is applied slowly, the PSEUDO-AUTO Mode is entered immediately. If the DIRECT RATE Mode is used initially, PSEUDO-AUTO Control begins when the Rate Error is less than the target Dead Band or when the time limit is exceeded. [5]

Included in the DAP design, are two Logic (Software) Components: an Attitude State Estimator, and the Reaction Control System (RCS) Control Laws (see figure 6.1).

The Attitude State Estimator derives the Angular Velocity and Angular Acceleration of the LM, based on measurements of LM Attitude and assumed Control Response only. Rate Gyros are not used. The basic measurements of the LM Attitude State available to the DAP are the Gimbal Angles of the (PGNS) Inertial Measurement Unit, which are sampled every (DAP Interrupt cycle: see above) 0.1 Sec. [26]

The TJETLAW section of the DAP code is the Automatic and ATTITUDE-HOLD Mode RCS Control Law. TJETLAW uses Phase Plane Logic about all the three LM Axes (P, U, V) to determine Command times and Firing duration for the RCS thrusters (Tjet). For a given Axis, the time between computations of Tjet may be either 0.1 Sec. or 0.2 Sec. The 0.2 Sec. interval results whenever a skip of the Tjet Computation has been requested. As described in the DIRECT RATE Mode section, a skip request is made for an Axis whenever Tjet has been determined to be less than 150Msec. (but not zero). If Tjet is either zero or greater than 150Msec., no skip is requested and a new Computation of Tjet is made in 0.1 sec. [6]

The TJETLAW Phase Plane Logic Switching Curves are Parabolic for the LM alone configuration. Attitude is controlled within an established Dead Band.

These Switching Curves are (for the LM alone configuration) shaped as a function of the Mass estimate. [6]

The simplest form of RCS Control Law would turn-ON and turn-OFF the RCS Jets only at the Control Sample instants (this is the way by which old 1950s or early 1960s Analog Rate-Command Systems were designed). Due to the DAP sample interval, and due to changes in LM Ascent Stage light-weight configuration, this RCS Control Law form is unacceptable. To overcome this difficulty, the assumed (computed, not measured in Real-Time) control effectiveness of the RCS Jets is used to determine the exact Jet firing durations that are required to deliver a desired change in angular velocity. [26]

DAP Assembly Language Source Code for the P-Error calculation and Dead Band Calling:

```

169  CALCPERR  CA  CDUY      P-ERROR calculation
170          EXTEND      CDU Counters are read in Two's Complement notation. Hence
171          MSUB  CDUYD    the MSUB instruction= {Measured CDU - Desired Angle (Y-Axis)}
172          EXTEND      (see page 1-40 on reference 30 for a description of MSUB)
173          MP   M11      Scaled at PI Radians
174          XCH  E        Save First Term
175          CA  CDUX
176          EXTEND      (see discution of MSUB on page 16)
177          MSUB  CDUXD    Measured CDU - Desired Angle (X-Axis)
178          EXTEND
179          MP   M13
180          AD  DELPEROR   Kalcmanu Interface Error
181          ADS E         Save Sum of terms
182          XCH  PERROR    For display in the FDAI
183          TC   Q         Return

```

```

*****
* Variables used in this section:
*
* OMEGAP ----- Estimated P-Axis Rate Scaled at PI/4
* AXISCTR ----- Axis indicated to the TJETLAW Subroutine
*****

```

```

184  PURGENCY  TC  CALCPERR  Calculate P-Axis Errors
185          CS  OMEGAPD     One's complement of Desired P-Axis Rate
186          AD  OMEGAP      Acumulator = (Estimated Rate - Desired Rate)
187          TS  EDOTP       EDOTP = (Estimated Rate - Desired Rate)
188          CS  ONE         Indicates the P-Axis in AXISCTR for TJETLAW
189          TS  AXISCTR
190          CA  DAPBOOLS
191          MASK CSMDOCKD
192          EXTEND
193          BZF  HEADTJET
194          INHINT          If CSMDOCKD = 1 Then Go To Docked RCS Logic
195          TC   IBNKCALL
          .....
196  HEADTJET  CA  ZERO
197          TS  SENSETYP
198          INHINT
199          TC  IBNKCALL
200          CADR TJETLAW
201          RELINT

```




Figure 6. Automatic RCS Control Law Phase Plane logic (reference 6)

As described before, TJETLAW subroutine is called when the LM is not Docked and the DAP is in the Automatic, ATTITUDE-HOLD or in the PSEUDO-AUTO Mode to Calculate the Jet-Firing-Time (T_{jet}) required for the Axis indicated by AXISCTR variable. The Subroutine is used in a peer-Axis Mode.

The Sign (+/-) of the required Rotation is carried out through TJETLAW as ROTSENSE. This Subroutine implements the RCS Phase Plane Logic shown in Figure 6.

```

*****
* Input Parameters:                                     *
* *                                                    *
* * AXISCTR --- -1 for P-Axis, +0 for U-Axis, +1 for V-Axis *
* * E ----- Attitude Error                          *
* * EDOT ----- Rate Error                           *
* * SENSETYP - Shows whether unbalanced couples are preferred *
*****
202   TJETLAW      EXTEND
203           QXCH  HOLDQ

```

DIRECT RATE/PSEUDO-AUTO Switching is carried out independently in the (LM) P-Axis and Q,R-Axes; thus it is possible to hold Attitude in YAW (P-Axis) while urgent Commands are issued in Pitch and Roll and vice versa. The P-Axis Coding (Program section) is separated from the Q,R-Axes Coding (Program sections) to allow a Manual Yaw maneuver during powered flight (see the manual Yaw maneuver done by the Apollo 11 crew during Powered Descent Phase).

The Transportation Lag, for the Rate-Command-Augmentation system, ranges from 130ms to 250ms. The AC Voltage output of the ACA feeds the Analog-to-Digital converter ("A" Circuit, plus LGC Counter system), which must be zeroed and enabled each time a reading is desired. The ACA Counters are not read until the detent switch is closed. Once this happens, there is a 100ms delay until the next DAP sampling instant. Although the Controller is deflected and is indicating a Voltage, the Counters must be zeroed and enabled, and the rate commands are zeroed for the present DAP cycle. On the next cycle, the counters contain valid (but 100ms old) data. There is an additional 30ms to 50ms computational and thrust-buildup delay. [6]




Figure 6.1. Manual Rate-Command-Augmentation control diagram (reference 6)

Rate-Command-Augmentation feature in Control Systems has been in use since mid-1950 and specially evaluated during the X-15 and Lunar Landing Research Vehicle projects (1959 - 1964). In some mid-1950s references this Control Mode was called "Control systems that Command Velocity" (see reference 7).

Here are two commentaries related to Rate-Command-Augmentation Mode uses in actual flights:

During Apollo 11 landing sequence, Commander Armstrong expressed the ACA testing procedure used (This transcription/commentary was extracted from the Apollo Lunar Surface Journal Web Page):

<Citation>

102:41:51 Armstrong (on-board):

" Okay. 5000 (feet altitude). 100 feet per second descent rate) is good. Going to check my attitude control. Attitude control is good "

102:41:51 Duke:

" Eagle, you're looking great. Coming up 9 minutes "
(Pause)

102:42:05 Armstrong:

" Manual attitude control is good "

[Commentary]

Armstrong - " We wanted a check that we were getting response from the hand controller before we were irretrievably committed to it. I think (that), like control-stick steering, You Could (intentionally) overpower the autopilot and, if you let go, it would go Back (to autopilot). We're in P64 and we just probably exercised the hand controller To check its response like control-stick steering, and then just released the Hand controller and it went back to the (computer) commanded attitude. "

<End of citation> Ref. [14]

In this Apollo 11 transcription/commentary, when Armstrong wrote about "Overpower" the Autopilot "as in Control-Stick-Steering", he is referring to his own experiences using the X-15 Command-Augmentation System.

This "Control-Stick-Steering" Mode was a Manual-Attitude-Guidance capability implemented in a Self-Adaptive Control system installed as a test bed in the X-15 No. 3 in 1961. By using this "Control-Stick-Steering" Mode, the Pilot could interrupt ("Overpower") the normal function of the Automatic Guidance System (The Autopilot implemented was an Analog Electronic Network at that time, and not a Digital entity) and move his Stick hand controller, establishing a new reference for the Attitude Control System (this is an Autopilot concept, and not a basic control concept). When the Stick hand controller is returned to the Detent position (neutral or center position), the Rate-Command-Augmentation part of the Self-Adaptive system takes action in order to null the vehicle Rates (following its Internal configured Dead Band Parameters), and then held the Space/Aircraft in the new desired (commanded) inertial position.

The Stick hand controller used in the X-15 Flight Control System was an Electrical Proportional controller on which a Strain-Gauge Bridge (an Analog Pressure transducer) sensed pilot force applied. The Strain-Gauge was built from a nylon-filled phenolic material. The Stick output was a +/- 40 Volt. Signal. The Stick Hand controller Electrical inputs (which were Proportional as in the Apollo LM system) were processed in the Following manner: The stick displacements are inserted in several Electronic Channels (only for Pitch and Roll axes). The Electrical input signal for each Channel was "shaped" by a simple Electronic Network, which had the dynamic response desired of the vehicle in that axis. These Networks were called "Models". [23]

The Self-Adaptive System and Command-Augmentation Mode was co-designed, built, and tested by Minneapolis-Honeywell in the period between 1958 and 1959 (reference 32). The system was called MH-96 and was the first Self-Adaptive system implemented in a high performance vehicle. Rate-Command-Augmentation was a "by-product" of the Self-Adaptive implementation.

NOTE:

The word "Overpower" is the same word used in X-15 technical papers, post-flight technical debriefings and Pilots Reports (see reference 32).

In the Apollo 17 Technical Debriefing, Commander Eugene Cernan said:

" Manual control of the spacecraft (the LM) was hard and firm, different certainly than the Command Module Operation but exactly what I expected the LM to be. The simulator (the LM Mission Simulator), I think, Does an excellent job of controlling the firm good solid ATTITUDE-HOLD, RATE-COMMAND capabilities of the LM " [16]

The Primary Control Path also includes two Discrete ACA processing Modes: the first one is called MINIMUM IMPULSE Mode and is implemented in the same DAP code of RATE-COMMAND-AUGMENTATION

Mode. The second is called LPD REDESIGNATOR and is embedded in the same program body of the Lunar Landing Guidance Equations. (See Luminary listing in Reference 5)

PGNS MINIMUM IMPULSE Mode

This Mode is available to the crew when a Verb 76 (V76E) is entered through the DSKY interface. This Mode provides a single 14-msec. thruster Pulse each time the ACA is moved Out-of-Detent; each Pulse results in an Angular Velocity increment whose Magnitude is a function of vehicle Inertia. [20]

Selection of the MINIMUM IMPULSE Mode enables the crew to perform an economical low Rate maneuver to a new orientation of the LM. After completing the maneuver, the crew codes Verb 77 (V77E) on the DSKY, and the DAP (see the RATE-COMMAND-AUGMENTATION section) returns to ATTITUDE-HOLD and causes the LM to limit cycle, with the normal Attitude Steering about the new orientation (requested). [27]

At the ACA Electrical Circuit level, when MINIMUM IMPULSE Mode is selected (V76E), a LGC Common 28 Volt. DC Excitation Source is applied to the Plus and Minus Switches of the ACA (both for each Axis), returning to the LGC Channel 31 Discrete Inbits when the ACA Handgrip is displaced more than 2.5 Deg in every Axis (compares this with the AGS PULSE Mode in the AGS Processing section).

ACA Switch Discrete Output (any Axis, and any sign) enters the LGC through a DC Interface Circuit called the "D Circuit" (see Figure 7). D Circuit receives the 28-Volt. DC ACA Switch Output and convert this Discrete to a level compatible to the LGC Micrologic Circuits. [4]

The MINIMUM IMPULSE Mode processing code is located near by the RATE-COMMAND processing code in the DAP section.

Channel 31 Inbits 1, 2, 3, 4, 5, and 6 receive the MINIMUM IMPULSE Discrete for PLUS and MINUS ACA Commands:

Bit 1 ----> Rotation Commanded in Positive PITCH direction
(Also Positive LPD Elevation Change)

Bit 2 ----> Rotation Commanded in Negative PITCH direction
(Also Negative LPD Elevation Change)

Bit 3 ----> Rotation Commanded in Positive YAW direction

Bit 4 ----> Rotation Commanded in Negative YAW direction

Bit 5 ----> Rotation Commanded in Positive ROLL direction
(Also Positive LPD Azimuth Change)

Bit 6 ----> Rotation Commanded in Negative ROLL direction
(Also Negative LPD Azimuth Change) Ref. [3] [5] [22]




Figure 7. The Minimum Impulse and Discrete LPD ACA Inteface ("D" Circuit at right)

For MINIMUM IMPULSE Mode processing, the DAP first check for the <"MODE CONTROL"> Switch Discrete to be in <"ATT HOLD"> position (input bit to the LGC). If this condition is Set, then the DAP proceed to check for the V76 or MINIMUM IMPULSE active condition.

DAP Assembly Language Source Code for "QRCONTRL":

NOTE:

This DAP Code section example is only for the Q and R Axes. P-Axis processing is similar.

```

204 QRCONTRL CA BIT13 Check for MODE SELECT Switch
205 EXTEND
206 RAND CHAN31 Bits inverted
207 CCS A
208 TCF ATTSTEER In AUTO Mode Control
209 CHKBIT10 CAF PULSES PULSES = 1 for MIN IMP use of ACA
210 MASK DAPBOOLS
211 EXTEND
212 BZF CHEKSTIK In Rate-Command/ATT-HOLD if bit10 = 0

```

```

*****
* MINIMUM IMPULSE Input Processing section *
*****
213      INHINT
214      TC      IBNKCALL
215      CADR    ZATTEROR
216      CA      ZERO
217      TS      QERROR
          ...
218      RELINT
219      EXTEND
220      READ    CHAN31
221      TS      TEMP31
222      CCS     OLDQRMIN
223      TCF     CHECKIN

*****
* Verify Inbits in Channel 31 (see Channel 31 bits above for reference) *
*****
224 FIREQR  CA      TEMP31
225         MASK    BIT1
226         EXTEND
227         BZF     +QMIN
228         CA      TEMP31
229         MASK    BIT5
230         EXTEND
231         BZF     +RMIN
232         CA      TEMP31
233         MASK    BIT6
234         EXTEND
235         BZF     -RMIN
236         TCF     XTRANS
237 CHECKIN  CS      TEMP31
238         MASK    OCT63
239         TS      OLDQRMIN
240         TCF     XTRANS

*****
* This is the Heart of the ACA MINIMUM IMPULSE Mode Processing *
*
* Variables and Constants used:
* 60MS -----> OCT140
* MINADR -----> GENADR MINRTN
* OCT63 -----> OCT 63
* 14MS -----> +TJMINT6
*****
241 +QMIN    CA      14MS
242         TS      TJU
243         CS      14MS
244         TCF     MINQR
245 -QMIN    CS      14MS
246         TS      TJU
247         CA      14MS
248         TCF     MINQR
249 +RMIN    CA      14MS
250         TCF     +2
251 -RMIN    CS      14MS
252         TS      TJU
253 MINQR    TS      TJV
254         CA      MINADR
255         TS      RETJADR
256         CA      ONE
257         TS      OLDQRMIN
258 MINRTN   TS      AXISCTR
259         CA      DAPBOOLS
260         MASK    CSMDOCKD
261         EXTEND

```

262	BZF	MIMRET	
263	INDEX	AXISCTR	If Docked, use 60MS MINIMUM IMPULSE
264	CCS	TJU	
265	CA	60MS	
266	TCF	+2	
267	CS	60MS	
268	INDEX	AXISCTR	
269	TS	TJU	
270	MINRET	CA	DAPBOOLS
271	MASK	AORBTRAN	
272	CCS	A	
273	CA	ONE	
274	AD	TWO	
275	TS	NUMBERT	
276	TCF	AFTERTJ	

A user's comment about the PULSE Mode came from Apollo 14 Commander Shepard. During Apollo 14 Technical Debriefing he said:

" After undocking I decided i was going to try to fly PULSE as much as i could to save gas...We had no control system problems, so most of the time I was flying PULSE. I had no problem in handling the LM in PGNS PULSE (that is, MINIMUM IMPULSE)...It handles real well that way, particularly with the heavy separation weight. Changes in rate and attitude can be very discrete and precise. So, undocking, formation flying, and tracking during the separation were no problem at all. " [24]

LPD REDESIGNATOR

Permits Man-in-the-Loop capability within the Lunar Landing Guidance Loop. Targeting is a key word here. Guidance is not a concern area on this monograph, but there is a very important ACA manual input, which takes place in this Lunar Landing Guidance process.

At the beginning of the Approach phase, the Automatic Landing System (Guidance Program P64) is targeted to a reselected Landing site. Prior to the approach phase, there is a completely automatic braking phase, which takes the LM from Lunar Orbit to those target conditions, which properly set up the approach phase.

The essence of the approach phase guidance system is that the Commander can Manually Steer the LM to the selected landing site, yet the trajectory he flies is produced by an automatic guidance system (Man-in-the-Guidance-Loop). This hybrid combination allows very nearly continuous Manual Steering in the form of commands, which incrementally shift the Point on the surface at

which the LM will land. The Commander must assess the safety of His trajectory. He must be assured impact is not imminent either because of equipment failure, or because of some gross Terrain obstruction along his course to the Landing Site.

Also, It probably will be necessary to change the direction of the trajectory in order to approach the Landing Site from a direction of more favorable lighting: to converge on a site judged adequate through detailed scrutiny and favorable lighting. At this respect, the Optimum Trajectory would approach the Landing Site nearly horizontally. However, the shallower The Trajectory, the less apparent are any shadows cast from the rear by the sun. Thus, the Depression Angle is a compromise Between Fuel Cost and Visibility. [17]

In the Commander's window there is a vertical Scale inscribed with numbers from 0 to 50 Deg. in vertical axis and from 0 to 10 Deg. in the horizontal axis (corresponding Elevation Deg. and Azimuth Deg.). This is so called Landing Point Designator or LPD.

At the beginning of the approach phase the LM assumes an attitude such that the surface and the (targeted) landing site are Visible, and the Commander visually scans the moon (surface) for the desired landing site. He will recognize it either by a marker placed by a previously landed Spacecraft, as a visually identifiable Landmark, or as an appropriate though unmarked Site. Meanwhile, the LGC orients the LM such that the (Descent Engine) thrust points in the direction

required for reaching the current site and uses the remaining degree of freedom (about the thrust axis) to keep the LPD (point in the reticle on the window) Superimposed on the current site. The LGC also repetitively calculates and displays, on Register 1 (R1) in the DSKY, The complement of the look angle (Deg. to search for in the windows scale) of the Current site. The Lunar Module Pilot or Systems Engineer repetitively reads the angle displayed on Register 1 (R1) and repeats it to the Commander. The Commander Identifies the Current site by sighting through the LPD window scale and observes (Man-in-the-Guidance-Loop) the Angular Error, if any, between the Current site and the Desired site. If the Angular Error is significant, he manipulates the ACA to cause The LGC to Redefine the Current Site closer to the Desired Site (see Figure 7). Through repetition of this process, the Commander literally steers (Man-in-the-Guidance-Loop) the Current Landing Site into coincidence with the Desired (Landing) Site. The Commander may steer in this way until about 15 Sec. prior to the end of this Approach Phase, when the LGC will stop accepting Steering Commands. [17]

Within the Lunar Landing Approach Phase Guidance Equations structure, the LPD processing is included as part of so called Manual Steering section.

Asked about the proposed LPD Redesignation capability, Eldon Hall wrote:

" As for the LPD idea, it was a method being considered to change the Landing site before the change to the digital autopilot. We just did not have a design to accomplish the requirement. So the method was developed and approved by NASA during the Implementation Meetings in 1964 which was the same time the digital autopilot was introduced. " [21]

The Manual Steering Equations must produce a current landing site in inertial coordinates and a display (LPD) number (in R1 on the DSKY) indicating the site. The new (Current) Landing Site is produced by the motion of the moon relative to the inertial coordinates, and by The Steering Commands (ACA Man-in-the-Guidance-Loop Commands). The Landing Site in inertial coordinates and the corresponding time are maintained as a consistent set. To determine the Site on the current pass (Guidance Processing take place iteratively with a time Interval of 2 Sec.), the Landing Site of the previous pass is rotated in the inertial coordinates by the angle traversed by the Moon Since the previous pass. Then the Site is corrected according to any Manual Steering Commands (ACA Man-in-the-Guidance-Loop Commands) Received during the interval, and stored along with the Current time for future reference. [17]

There is no way to synchronize the Manual (Steering) Commands (ACA Man-in-the-Guidance-Loop Commands) with passes (iterations) through The Guidance Equations. Consequently, it is necessary to interrupt the (Lunar Landing Approach Phase program) execution when a (Manual) Steering Command is received and increment or decrement the Azimuth or Elevation Command Count. On the coming pass, the Command Counts are used by the Landing Site Perturbation Equations (which are part of the Lunar Landing Approach Phase Guidance Equations structure). [17]

The LPD Redesignator processing is started by a Program Interrupt within the LGC Sequence Controller. The Interrupt is started any time by the ACA movement.

Within the LGC, each Program Interrupt is mechanized through the Involuntary Instruction RUPT.

The Hand Controller Interrupt (Interrupt 10) is generated by the Out-Of-Detent Switches in the ACA (see ACA Electric Source and Electric Output section above) and is detected from bits Of Channel 31 and 32. The memory address for the Interrupt processing is 4050 octal Address. [4]

Following the program sequence in the Luminary Code, the Interrupt 10 could take place at every moment and during every Major Mode. But during the Lunar Landing Approach Phase Guidance, the "PITFALL" Trap Handler differentiate the LPD Discrete Input processing from the Real Time ACA processing within the Digital Autopilot (DAP) by checking for the Switch Discrete ATTITUDE-HOLD present, doing so the Trap Handling (not a common ACA Interrupt Handling).

NOTE:

There are two different Interrupts on which the ACA Processing is involved. One of them is Interrupt 5 for DAP Processing. The other one is the Interrupt 10 described in this subsection. The Control Logic to differentiate both ACA Processing Paths is established by the ATTITUDE-HOLD Mode Control Switch Discrete. As we will see in the Source Code below, DAP Control Logic (10 times per-sec.) check for this Discrete Inbit, and if it is in ATTITUDE-HOLD Mode, the LPD Guidance Loop Processing is "Interrupted" (including the

Interrupt 10 Control Logic).




Figure 7.1. LPD Window Reticle, ACA Command Axes, and PGNC S Mode Control Switch

At the ACA Electrical Circuit level, when the ACA LPD capability is available, the same MINIMUM IMPULSE positive OR Negative DC Excitation Voltages applied to the Plus and Minus Switches of the ACA return to the LGC common when the ACA Handgrip is displaced more than 2.5 Deg in every Axis.

Assembly Source Code for "REDESIGNATOR INTERRUPT LEAD INSTRUCTIONS" (Interrupt 10 Trap Processing at address 4050):

Fixed-Address	Instruction
4050	DXCH ARUPT
4051	CA RUPT10BB
4052	XCH BBANK
4053	TCF PITFALL Transfer control to the Redesignator Trap code

Assembly Source Code for "REDESIGNATOR TRAP":

277	PITFALL	XCH	BANKRUPT	
278		EXTEND		
279		QXCH	QRUPT	
280		TC	CHECKMM	If not in P64, no reason to continue
281		DEC	64	
282		TCF	RESUME	
283		EXTEND		
284		READ	CHAN31	
285		COM		
286		MASK	ALL4BITS	
287		TS	ELVIRA	
288		CAF	TWO	
289		TS	ZERLINA	
290		CAF	FIVE	
291		TC	TWIDDLE	
292		ADRES	REDESMMON	Redesignator Monitor Memory Address

```

293          TCF  RESUME
          . . . .
294  PREMON1  TS    ZERLINA
295  PREMON2  CAF  SEVEN
296          TC    VARDELAY
297  REDESMON EXTEND
298          READ  31
299          COM
300          MASK  ALL4BITS
301          XCH  ELVIRA
302          TS    L
303          CCS  ELVIRA      Do any bits appears this pass?
304          TCF  PREMON2    Y: continue Monitor
305          CCS  L          N: Any last pass?
306          TCF  COUNT'EM   Y: COUNT'EM, reset RUPT, Terminate
307          CCS  ZERLINA    N: has ZERLINA reached zero yet?
308          TCF  PREMON1   N: Diminish ZERLINA, continue
309  RESETRPT TC    C13STALL Y: Reset RUPT, terminate
310          CAF  BIT12
311          EXTEND
312          WOR  CHAN13
313          TCF  TASKOVER

*****
* At this point, the program check for ATTITUDE-HOLD Switch Mode. See note above. *
* *
* ATTITUDE-HOLD is not a part of the Guidance Loop. It is a Control Mode. Hence, *
* while in ATTITUDE-HOLD, any guidance stop and vehicle Attitude is kept. *
* *
* During P64 Major Mode execution, the Commander can toggle the PGNS Mode Control *
* Switch from "AUTO" to "ATT HOLD", getting the RATE-COMMAND/ATTITUDE-HOLD *
* capability And taking over the LM control over the three axes, maybe to check the *
* ACA response before the crew is irretrievably committed to it". *
* (See Commander Armstrong comments on the Apollo 11 landing phase above). *
*****
314  COUNT'EM  CAF  BIT13      Are we in ATTITUDE-HOLD
315          EXTEND
316          RAND  CHAN31
317          EXTEND
318          BZF  RESETRPT    Yes: skip REDESIGNATON LOGIC

*****
* Start to review the LPD Redesignation Discrete Input bits *
* -Azimuth, +Azimuth, -Elevation, +Elevation *
* This is the Heart of the ACA LPD Input Processing: *
* *
* Variables and Constants used: *
* +ELBIT -----> Bit 2      -Pitch (Channel 31) *
* -ELBIT -----> Bit 1      +PITCH (Channel 31) *
* +AZBIT -----> Bit 5      (Channel 31) *
* -AZBIT -----> Bit 6      (Channel 31) *
* ALL4BITS -----> 00063 (Octal) *
* AZEACH -----> .03491     2 Deg. *
* ELEACH -----> .00873     1/2 Deg. *
*****
319          CA    L          No
320          MASK  -AZBIT
321          CCS  A
322  -AZ      CS    AZEACH
323          ADS  AZINCR1
324          CA    L
325          MASK  +AZBIT
326          CCS  A
327  +AZ      CA    AZEACH
328          ADS  AZINCR1
329          CA    L
330          MASK  -ELBIT

```

```

331         CCS    A
332  -EL      CS    ELEACH
333         ADS    ELINCR1
334         CA     L
335         MASK   +ELBIT
336         CCS    A
337  +EL      CA    ELEACH
338         ADS    ELINCR1
339         TCF    RESETRPT

```

NOTE:

According to Professor Andrew Tanenbaum in reference 18, A Trap is:


" A kind of automatic procedure call initiated by some condition caused by the program, usually an important but rarely occurring condition...the flow of control is switched to some fixed memory location instead of continuing in sequence. At that fixed location is a branch to a procedure called the Trap Handler, which performs some appropriate action. " [18]

About the definition of Interrupt, reference 18 describes:

" Interrupts are changes in the flow of control caused not by the running program, but by something else, usually related to I/O...The essential difference between Traps and Interrupts is this: Traps are Synchronous with the program and Interrupts are asynchronous. " [18]

In reference 19, Dr. Donald Knuth makes an interesting statement about the Trap topic:

" MMIX distinguishes two kinds of program interruptions: Trips" and Traps. A Trip sends control to a Trip Handler, which is part of the User's program; a Trap sends control to a Trap Handler, which is part of the Operating System. " [19]



2. AGS PROCESSING (Secondary Control Path)

Through the backup attitude control path, the crew can use one of four Control Modes for Attitude maneuvering. Each Mode could be selected in an individual or pear-axis way:

- AGS RATE-COMMAND/ATTITUDE-HOLD Mode

(<"ATTITUDE CONTROL"> Switch in <"MODE CONT"> and <"MODE CONTROL"> Switch in <"ATT HOLD">)

The ACA Electric Proportional output (as described in Sections 0. and 1.) enters the ATCA for processing. The ATTITUDE-HOLD feature engages whenever the ACA Stick is returned to the Detent position.

While engaged in the ATTITUDE-HOLD Mode, Rate Damping is achieved (inside the ATCA) by summing the Rate Command with the Rate Gyro feedback Signals (see Figure 11). [27]

ACA Proportional Output shall be active through the entire ACA deflections (+/- 10 Deg.) except for the Dead Band (see the ATCA Processing below) at neutral. This is a Mechanical/Electrical Dead Band, which do not exceed a +/- 1 Deg. ACA deflection. The Proportional Output falls outside the Mechanical Dead Band to prevent inadvertent inputs. [12]

- AGS PULSE Mode

(<"ATTITUDE CONTROL"> Switch in <"PULSE"> position)

This is an Open-Loop mode on which the crew can obtain a single impulse by moving the ACA Handgrip to 2.5 Deg. Deflection or beyond and then returning it to detent position. The Electrical Source for ACA AGS PULSE Mode Commands is generated in the ATCA DC excitation source.

With Attitude Control Switch placed in "PULSE" for a given axis, positive OR negative DC voltages from the ATCA are applied to the Plus and Minus Switches of the ACA. So, when the ACA Stick is displaced 2.5 Deg. Or more from neutral, a positive or negative DC voltage is applied to the ATCA in return. In this Mode the Detent Switch output go directly to the Secondary Coils of the reaction jets.

Opposed to the single Jet firing commanded while in the PGNS MINIMUM IMPULSE Mode, in the AGS PULSE Mode provides a series of Minimum Impulse Jet firings when the ACA is deflected. [27]

The PULSE Mode is actuated at the same Handgrip displacement point as the DIRECT Mode section (see bellow).

- DIRECT Mode

(<"ATTITUDE CONTROL"> Switch in <"DIR"> position)

See DIRECT MODE (2-Jets Path) section bellow.

- EMERGENCY DIRECT Mode

(4-Jets)

See section 4, DIRECT MODE (Emergency Path).

AGS Processing follows the technological heritage from Gemini Spacecraft program on the use of Transistor Switching Logic for Electric Control of the Spacecraft using Analog Circuitry (not Digital). Grumman's original Specification for the Lunar Module Primary Control system was to implement it as an Analog Electronic Network. The result of this initial design and arrangement can be seen on the Abort Guidance System or AGS: its data input and readout terminal (called DEDA) is located in the Lunar Module Pilot position at right, as the in Gemini Spacecraft panel.

Multiple weight limitations on the Lunar Module design stimulated the use of Software to implement the Primary Control functions using the LGC as an Input/Output Programmable Array, and

leaving the original Analog Electronic design as a Backup secondary system available at every time to take over LM control.

Unlike the Apollo Command Module Flight Control System, the LM design was influenced from the very beginning by Pilot's opinion. Charles Conrad, Tomas Stafford, and Neil Armstrong were some of the various Pilots who were called to put their experience on Electronic Switching Logic from Gemini and from other Command-Augmentation and Direct-Acceleration implementations during the late 1950 and early 1960.

Referring to the LM original control system and the proposition for use a Digital Autopilot, Cline Frasier from NASA Guidance and Control Division, wrote:

" The control system of the LM was near and dear to the hearts of the astronauts. Fortunately (for the Digital Autopilot idea), all was overcome by LM weight problems, some very good DAP work by George Cherry, Bill Widnall and others at MIT/IL, and the precedence of the CM (Apollo Command Module) system. If we had tackled Grumman and the LM first, I doubt that we would have had a DAP. " [11]




Figure 8. ACA and Stabilization/Control Switching Panel

ATTITUDE AND TRANSLATION CONTROL ASSEMBLY (ATCA)

At the heart of the AGS Control Path Processing is the Attitude and Translation Control Assembly or ATCA. This "control box" contains all the Circuitry that ground to Analog Electronic Circuitry the AGS Control Path. This Electronic Circuitry is the counterpart to the DAP Software entity described in Section 1.

The ATCA is composed of four Output Subassemblies, three Analog Subassemblies, one Power Supply, one Wiring Subassembly, the Assembly Chassis and Cover, Electrical Connectors, and one Elapsed-Time indicator.

The ATCA provides control of LM rotation in all three axes, through three channels (one Channel per axis). It provides the same basic Autopilot Modes provided by the DAP Software in the Primary Control Path described in Section 1; that is, Automatic, Rate-Command/Attitude-Hold, and Pulse. In these Modes, the ATCA (rather than the LGC) accepts Signals from the ACA, from the Rate Gyro Assembly (RGA), and provides an Error Signal that manage the necessary Jet select Logic for Jet firing. [27]

Within the ATCA Proportional Circuitry the ACA Input is Scaled to a Maximum Commanded Rate (MCR) of 20 Deg./Sec. about each one Axis. This is the NORMAL ACA Scaling established. [2] [15]

NOTE:

The RGA senses pitch, roll, and yaw rates and applies 800 cps rate signals to the ATCA. The ATCA uses the 800 cps rate signals to rate-damp the attitude control loop when AGS control is selected. Rate Signals are displayed on the FDAI (for crew visual feedback) from monitor points at the ATCA inputs.

The RGA consists of three single-degree-of-freedom rate gyros mounted so hat they sense vehicle roll, Pitch, and yaw rates. Each gyro can measure input rates from -25 Deg./Sec. to +25 Deg./Sec. Each Rate gyro senses a rate of turn about its input axis. An Analog output signal (rate of turn) is Obtained from a stator-rotor Pickoff, which is positioned by the precession of the gimbal. The Pickoff Voltages are CONTINUOUSLY fed to the FDAI rate indicators and to the ATCA.

Each rate Gyro is a conventional aircraft-type Rate Gyro modified to accept 800 cps power for compatibility with the PGNS power frequency. [27]

Three-phase, 26 volt, 800cps power generated in the ATCA is applied to the motor of each gyro. Pickoff Excitation (28 volt, 800 cps power generated in the ATCA) is applied to the primary coil of a microsyn Pickoff. The secondary coil picks off an 800 cps voltage proportional to the precession of the gyro, Hence, proportional to the LM rate about the gyro input axis. Polarity is arranged so that an in-phase Pickoff voltage indicates a positive rate, and a 180 Deg. phase Pickoff voltage, a negative rate. [3]




Figure 9. The Attitude and Translation Control Assembly (ATCA) Interfaces and Power Supply

ATCA Rate Dead Band management is possible by instructing the ATCA in two Dead Band Modes: MAX and MIN. Selection of the Dead Band Mode is done by way of the <"DEAD BAND"> Switch located in the Stabilization/Control Switching Panel. There are two positions labeled <"MAX"> and <"MIN"> (See Figure 8).

When the <"DEAD BAND"> Switch is set to <"MIN">, the Error Signals (from ACA and RGA Signals Summation) are first passed through a limiter to prevent excessive vehicle rates. The limiter maintains the vehicle rates at a maximum of 10 Deg./Sec. in Pitch and 5 Deg./Sec. in Roll and Yaw. When the <"DEAD BAND"> Switch is set to <"MAX">, the limiter is eliminated.

In resume:

MIN Dead Band is used for tight Rate Control using an electronic LIMITER module within the ATCA. The Dead Band is established within - 0.4 Deg. and + 0.4 Deg. One LIMITER by each Channel (Axis).

MAX Dead Band is used for wide Rate Control by eliminate the LIMITER module in the internal ATCA Signal Path. The Dead Band is established within -5 Deg. and + 5 Deg. The MAX Dead Band is used for coasting-period limit cycling to conserve fuel.

NOTE:

At the Power Source level, the ATCA power supply receives 28-Volt DC from the LMP's bus through the <<"STAB/CONT ATCA"> circuit breaker.




Figure 10. Gemini Attitude Control Electronics Proportional Circuitry (arrow indicates the Rate Summation point)

The ATCA Rate Signal Summation (Proportional Circuitry), is performed by the Rate Signal Summation Amplifier (see Figure 11). Inside the Rate Signal Summation Amplifier, the ACA and RGA Signals are "AC Summed" and then Amplified before demodulation into DC. This circuit level AC Summation Process is done in the same way as in the Gemini Spacecraft system (see Figure 10). Only the electronic component technology (implementation) is different. [29]




Figure 11. Apollo LM ATCA (arrow indicates the Rate Summation point)

In the Apollo 17 (an improved version of the Apollo 9 LM-3 RATE-COMMAND Mode) Technical Debriefing, Commander Eugene Cernan said:

" I checked it out (the AGS Path) in both, PULSE initially. I checked all three axes out in PULSE. I got the continuous rapid fire pulses. It checked out in three axes. I checked it out in RATE-COMMAND both for COMMAND and ATTITUDE-HOLD. And it was a very tight system. I checked it out in MIN DEAD-BAND only. It was GO. There was absolutely nothing wrong with the AGS system either during Power up or during the phases of checkout. " [16]

Switching logic used in AGS control mode (similar to the all electric Gemini), as opposed to digital Software configuration, depends on physical switching procedures, and it needs very good check-list Management as evidenced by the staging procedure during Apollo 10 mission. Staging was preceded by normal ascent batteries in line, so ascent stage electric power was used for the remainder of the mission. AGS Mode Control was in PULSE for the three axes. The DAP was reset (reconfigured) for the lightweight Ascent stage future configuration.

From the Apollo 10 Mission Report:

" Preflight planning required that the AGS control mode be used for staging and that, at 2 minutes before staging, the Mode Control Switch be placed in ATTITUDE HOLD and the Attitude Switches in MODE CONTROL. At approximately 5 seconds before staging, the LM started a motion that was characterized by a rapid roll rate accompanied by small yaw and pitch rates. The vehicle was staged with the planned velocity change of approximately 2ft/sec. In this situation, an attempt was made using the DIRECT Coils of the RCS to pitch the vehicle to avoid gimbal lock and to damp the rates. " [10]

The mode switching, telemetry, and associated attitude commands indicated that the AGS mode changed from ATTITUDE HOLD to AUTO coincident with the vehicle gyrations problem described above: a Switching Logic problem.

Recorded from Apollo 10 mission:

" Approximately 4 minutes before staging, with the guidance select switch in AGS and the attitude control switches in PULSE, the crew verified that the AGS mode control switch was in ATTITUDE HOLD since the intent was to perform staging in AGS ATTITUDE HOLD. After some discussion, they selected MAX DEADBAND to save propellant (in MAX, the phase-plane logic is very wide, preventing the RCS to start firing for fight against little rate deviations).

The AGS steering logic was set to Z-axis steering throughout the staging sequence. If AUTO mode is selected, Z-axis logic will produce the steering commands required to point the LM Z-axis at the Command Module. If the guidance select Switch is in AGS and attitude control switches in MODE CONTROL, the Z-axis steering commands are accepted and acted upon by the control system.

The attitude control switches were sequentially thrown to MODE CONTROL 51 seconds before staging.

Five seconds after the selection of MODE CONTROL, the mode Control switch indication changed from ATTITUDE HOLD to AUTO, as recorded, and remained in AUTO for 3 seconds then returned to ATTITUDE HOLD. During this period the LM moved in all three axes in response to the Z-axis steering commands. After the mode indication returned to ATTITUDE HOLD, the dynamics again returned to normal for WIDE DEADBAND. After approximately 40 seconds, the mode indication again returned to AUTO, and the LM responded to Z-axis steering commands. The LM was staged 4 seconds later, and the dynamic response increased abruptly. Because of the relative scaling of the ACA, Rate Gyros, and Attitude Errors, attempts to manually control the motion were ineffective, and the LM stabilize with the Z-axis pointing toward the Command Module. " [10]

ACA DIRECT control MODE was used at the end of the emergency Procedure. See the DIRECT MODE section below.

3. DIRECT MODE (2-Jets Path)

This Mode can be selected per-Axis either in PGNS processing (Primary Control Path) or in AGS processing (Secondary Control Path). The 2-Jets Path is active and usable in the following Guidance conditions:

- While <"GUID CONT"> Switch in <"PGNS"> position, and <"PGNS MODE CONTROL"> Switch in <"OFF"> position.
- While <"GUID CONT"> Switch in <"AGS"> position, and <"AGS MODE CONTROL"> Switch in <"OFF"> position.
- While <"GUID CONT"> Switch in <"PGNS"> position, and <"PGNS MODE CONTROL"> Switch either in <"AUTO"> or in <"ATT HOLD"> positions.
- While <"GUID CONT"> Switch in <"AGS"> position, and <"AGS MODE CONTROL"> Switch either in <"AUTO"> or in <"ATT HOLD"> positions.

This mode is entered whenever any of the <"ATTITUDE CONTROL"> Panel Switches is put (independently) in the <"DIR"> position. Direct connection to the Secondary Coils of the RCS Jets is made through the ACA Soft-Stop Switches.

The 2-Jets DIRECT is another Open-Loop Mode. As in AGS PULSE mode, the DIRECT mode operates through the 2.5 Deg. (25 percent of) ACA Handgrip deflection point. However, in this Mode the Detent Switch output go directly to the Secondary Coils of the reaction jets and fire two jets continuously for as long as the ACA is out of detent.

At the Power Distribution level the ACA receives 28-Volt. DC power from CDR's bus through the <"STAB/CONT"> <"ATT DIR CONT"> circuit breaker (see the Emergency Path DIRECT Mode bellow).

NOTE:

With no Guidance Control Active (PGNS or AGS), DIRECT Modes (2-Jets and/or Emergency) are the only way to command Manual Attitude Control.



Figure 12. Apollo X Mission Simulation showing Switching Panel Manipulation and ACA

4. DIRECT MODE (Emergency 4-Jets Path)

The final Attitude Control Mode is an EMERGENCY DIRECT, available in either the PGNS processing (Primary Control Path) or AGS processing (Secondary Control Path), even with no Guidance Mode active (see NOTE in 2-Jet DIRECT Mode above).

This mode is actuated whenever the ACA is displaced against the Hard-Stop angle position. Direct Connection to the Secondary Coils of the RCS Jets is made through the ACA Soft-Stop Switches.

The form of Acceleration Control employed in this Mode and in the 2-Jets Path, is an Open-Loop System in which only the direction of Angular Acceleration is commanded. The magnitude of Acceleration is established by the relation of the fixed Attitude Rocket Thrust Level to the moment of Inertia about the commanded Axis.

ACA DIRECT control is set available by close the <"ATT DIR CONT"> Circuit Breaker (<"STAB/CONT"> section), and then toggle to <"ENABLE"> position the <"ACA/4 JET"> switch. [3]

This EMERGENCY DIRECT mode is actuated at a point beyond the Soft Stops and before contacting the Hard Stops. [12]

Some technical literatures call this Mode as a "Hardover Mode" (see reference 3).

REFERENCES AND SELECTED BIBLIOGRAPHY

- [1] Transducers and Their Elements - Alexander D. Khazan - Prentice Hall PTR - 1994.
- [2] Handling Qualities for Pilot Control of Apollo Lunar-Landing Spacecraft - Cheatham, D. C., Hackler, C. T. - Journal of Spacecraft and Rockets, Vol. 3, No. 5, May 1966, pp. 632-638.
- [3] Apollo Operations Handbook - Lunar Module - LM 10 and Subsequent - Volume 1 - Subsystems Data.
- [4] MIT's Role in Project Apollo - Volume III - Computer Subsystem - Hall, Eldon C. - August 1972.
- [5] Luminary 10 - Source Listing - Section "Manual Rate Command Mode" - 1971.
- [6] Manual Attitude Control of the Lunar Module - Stengel, Robert F. - AIAA paper 69-892 - August 1969.
- [7] Simulator Investigation of Command Reaction Controls - Holleman, Euclid C.; Stillwell, Wendell H. - NACA High Speed Flight Station - July 17, 1958.
- [8] Personal feedback letters from Robert Stengel - between July 2001 and May 2003.
- [9] Apollo 9 Onboard Voice Transcription - recorded on the Lunar Module onboard Recorder Data Storage Equipment Assembly (DSEA) - March 1969.
- [10] Apollo 10 Mission Report - 1969.
- [11] Personal feedback letters from Cline Frasier - between July 1999 and August 2003.
- [12] Apollo Experience Report - Crew Station Integration - Volume III - Spacecraft Hand Controller Development - Wittler, rank E. - Lyndon B. Johnson Space Center, Huston Texas 77058 - March 1975.
- [13] Development of the X-15 Self-Adaptive Flight-Control System - Johannes, Robert P.; Armstrong, Neil A.; Hays, Thomas C. - January 1961.
- [14] Apollo Lunar Surface Journal - Web Page - 1999.
- [15] Apollo 9 Mission Report - 1969.

- [16] Apollo 17 Technical Crew Debriefing - January 4, 1973.
- [17] A Manual Retargeted Automatic Landing System for the Lunar Module - Klumpp, Allan R. - Journal of Spacecraft and Rockets - Volume 5 - Number 2 - February 1968 - pp 129-138.
- [18] Structured Computer Organization - Tanenbaum, Andrew S. - 4th Edition - 1999 - pp 379-380.
- [19] The Art of Computer Programming, Fascicle 1, MMIX - Knuth, Donald E. - 2001 - pp 18.
- [20] MIT's Role in Project Apollo - Volume V - The Software Effort - Johnson, Madeline S.; Giller, Donald R. - March 1971.
- [21] Personal feedback letters from Eldon C. Hall - between June 1997 and November 2003.
- [22] Interface Control Document - ICD No LIS-370-10004 Rev. D. - Grumman Aircraft Engineering Corporation Bethpage, L. I. NY - Initial Release January, 1965.
- [23] Personal feedback letters from Gary Hartmann - June 2002.
- [24] Apollo 14 Technical Crew Debriefing - February 17, 1971.
- [25] Computer Organization and Design - Patterson, David A.; Hennessy, John L. - 2nd Ed. - 1997 - pp 223.
- [26] Lunar Module Digital Autopilot - Widnall, William S. - Journal of Spacecraft - Vol. 8, No. 1 - January 1971.
- [27] Apollo Experience Report Guidance and Control Systems: Lunar Module Stabilization and Control System - Shelton, D. Harold - Report. No. NASA TN D-8086 - November 1975.
- [28] Introduction to Automatic Controls, Second Edition - Harrison, Howard L.; Bollinger, John G. - 1966.
- [29] Project Gemini Supplement Familiarization Manual - SEDR300 - McDonnell - MAC Control No. C-119162 - September 1965.
- [30] AGC4 Basic Training Manual - Volume I - Savage, Bernard I.; Drake, Alice - January 1967.
- [31] Logical Description for the Apollo Guidance Computer - Hopkins, Albert; Alonso, Ramon; Blair-Smith, Hugh - 1963.
- [32] Contributions of the X-15 Program to Lifting Entry Technology - Holleman, Euclid C.; Adkins, Elmor J. - AIAA paper 64-17 - January 1964.
- [33] Personal feedback letters from Hugh Blair-Smith - 2003.
- [34] Interim Report on Flight Evaluations of Lunar Landing Vehicle Attitude Control Systems - Hewes, Donald E. - AIAA paper 67-239 - February, 1967.

APPENDIX

APPENDIX A

A DISCUSSION WITH HUGH BLAIR-SMITH
ON THE APOLLO GUIDANCE COMPUTER SEQUENCE GENERATOR
AND
MICROARCHITECTURE [33]

APPENDIX A, INTRO NOTES

instructions had to have multiple purposes; in that connection see my article in the Gordon Bell referenced in my notes.

- 2) Today's Micrologic designers use more standard tools to implement a micro-program: PLCs or PLAs, ROM, etc. That is, whatever standard logic in opposite to more traditional "from scratch" sequential-combinational logic. What were your tools for AGC sequencer logic design?

HB-S: Primarily, a repertoire of control pulses that had been refined from Mod 1A through Mod 3S, and even owing something to the Harvard Mark IV and a "crosspoint matrix" to generate them in appropriate sequences. I think this matrix may fairly be called a forerunner of the PLA as we know it today. But it was implemented with 3-input NOR gates like everything else.

- 3) Was your Yul system requested for the production of punched cards for the direction of the Gardner-Denver wire-wrap machines in Raytheon?

HB-S: My friend Bob Morse wrote a program called Delphi (the oracle of Apollo) to produce those cards to drive the G-D machines, which were used to interconnect the logic modules. The modules themselves were built by Raytheon technicians using spot welding and other techniques. I don't know enough to add to Eldon's book on that point.

The Yul system did produce cards which were converted to a "paper" tape form for driving the rope-weaving machine mentioned in Eldon's book and my notes. Later, Yul produced the tape

directly. The cards, or at any rate their images on the disk file, were used by the simulator to test the programs.

APPENDIX A, SECOND SECTION

JP: I had been following the AGC processor logic design as you stated in your report R-393 (see reference 31), and in your email response to my first questions. In your letter response, you wrote:

" There weren't any of the formal named methodologies then; it was just a matter of what would work and make most efficient use of the resources... "

Following my review of your design report on AGC4, and the statement above here are other questions:

(A) On The SEQUENCE GENERATOR

1. In your report, page 2-9, you wrote:

" A control pulse is generated as the logical sum of a subset of the matrix crosspoints. A crosspoint, in turn, is the logical product of a subinstruction variable and a time pulse... "

This could be done by sum of products (Canonical form) with one first array of ANDs followed by another array of OR gates (outputs).

AGC used only NOR gates for all AND, OR and NOT logical operations, so:

How did you organize all this minterms (Canonical form) array with only NOR gates?

HB-S: When required, we could always use another NOR gate to produce the complement of any signal, but generally, we just arranged things so that such wasteful practices weren't necessary. For example, many subinstructions have the control pulses RZ WS ST2 occurring at time 8 (to start the fetch cycle for the next instruction).

Taking the subinstruction AD0 as a particular case, we can form a crosspoint for that line

with just one NOR gate if we have the complements of the time pulses and the subinstruction signals. Here, the NOR of T08/ and AD0/ is the crosspoint, which I'll call AD08 because I don't remember its real name. (We used a trailing slash as our notation for a complement signal: T08/ is at high voltage whenever it is not time 8.)

Then the complement signal RZ/ is the NOR of all the crosspoints that include an RZ pulse - AD08, CA08, etc., and the same is true of WS and ST2 in our example.

We had a way of ganging 3-input NOR gates together to make 6-input, 9-input (etc.) NOR gates. So RZ/ would be high (true) whenever all its incoming crosspoint signals were low, and low (false) whenever any of its crosspoints was high.

I believe that the bus wires were the NOR of all the signals from registers that could be read, which meant that RZ/ was the correct polarity to inhibit Z register bits from affecting the bus wire values.

Anyway, that's the point: by twisting our minds around to think NOR-fashion, we could usually get what we wanted with a minimum of gates.

That's how the canonical form worked (as I remember it), but of course there were fan-out and fan-in limits, and we had to use extra gates to get around those. And having designed the sequence generator as canonically as possible, I believe we then reduced the gate count by judicious complications of the logic -- in this example, we may have constructed a signal representing the whole RZ-WS-ST2 combination because that occurs in so many places. That's where most of the 'spaghetti' came from.

2. How you tested the timing pulses input and their combination with OP CODE bits? by oscilloscope maybe as today designers?

HB-S: I think we did use multi-trace oscilloscopes, but I wasn't involved in the hardware verification. My part was all done by desk-checking -- mental modeling of what the stated logic does.

(B) On the CENTRAL ADDER (eq. modern ALU)

1. Same question as in Sequence Generator part: an today's ALU is formed by several standard Full one bit Adders with previous level carry input, structured in cascade of full single bit adders. How do you structured all this NOR spaghetti? or was it done using a different approach than the one bit full adder cascade?
2. In the report, you only wrote about the X, Y input registers, but did not write about the arithmetic logic array in it-self. That's what i want to know in connection with question 1 above.

HB-S: This was quite interesting, because it's one of the places where the design improved greatly from model to model. In Block II, to support the quick additions required in multiplication, Al Hopkins designed an "implicit-carry" adder, meaning that there was a large tree of logic to handle all combinations of carries in the one pulse time.

To put it another way, instead of a cascade of 16 single-bit adders, we had one 16-bit adder: each bit of the result was unstable until all the rippling of add-and-carry logic had settled out. The NOR-gate implementation was achieved by the same sort of NOR-thinking that I described for the sequence generator, above.

Somewhere, there must be a gate connection diagram of the whole adder -- I sure hope so.

APPENDIX A, THIRD SECTION

JP: I reviewed your response last night. One question about your last response:

In your last email you wrote:

" Taking the subinstruction AD0 as a particular case, we can form crosspoint for that line with just one NOR gate if we have the complements of the time pulses and the subinstruction signals. Here, the NOR of T08/ and AD0/ is the crosspoint... "

My question is:

what do you mean on AD0: do you mean one single bit? or do you mean a complete opcode (3 bits on AGC)? if you are getting the complete AD opcode, then you are using a complete NOR (three inputs) micrologic gate for the opcode bits.

HB-S: yes, I meant a single bit produced by logic that resolves the op code into separate signals for each subinstruction. I'm fairly sure that subinstruction signals were in fact latched in a pair of cross-connected NOR gates for the (usually) 12 pulse times they were active; time 12 was dedicated to resetting all such latches and setting the one for the next subinstruction. The nice thing about those latches is that they generated both direct and complement signals, in this case, the one latch produced both AD0 and AD0/. Another latch produced both STD2 and STD2/, and so on.

GLOSSARY

ACA -----: Attitude Controller Assembly

RGA -----: Rate Gyro Assembly

AGC -----: Apollo Guidance Computer

DSKY -----: Display and Keyboard, this is the Input/Output Interface with the AGC

DAP -----: Digital Autopilot

FDAI -----: Flight Director Attitude Indicator

Luminary ----: Lunar Module Software (Includes the DAP)

AM -----: Amplitude Modulation

LM -----: Lunar Module

LGC -----: LM Guidance Computer (same as AGC)

Rate -----: Derivate of Velocity respect to time (Vd/Td)

IMU -----: Inertial Measurement Unit

ATCA -----: Attitude and Translation Control Assembly

RCS -----: Reaction Control System

PSA -----: Power and Servo Assembly

PGNCS -----: Primary Guidance, Navigation and Control System (as Software in the LGC)

RVDT -----: Rotary Variable Differential Transformer (Synchro), this sensing device Contains coils. The change in the magnetic circuit of the transduction Element due to the motion of the sensing shaft provides a change in the Mutual inductance between the coils. As a result, a voltage "Proportional" to the change in the circuit is induced in the output coil. The RVDT is a linear output transducer

Proportional : Two related quantities "x" and "y" are called "Proportional" if there exists a constant non-zero number "k" such that: $y = k*x$ in this case, "k" is called the Proportionality Constant of the relation

Velocity ----: Is the derivative (with respect to time) of an object's position

Acceleration : Is the derivative (with respect to time) of an object's velocity

RMS -----: (Root Mean Square) Math formula for calculating the value of an AC Signal